
Abstract. We present modular formalizations of the model specification
language Reactive Modules and the temporal logic CTL∗ in the proof as-
sistant Coq. In our formalizations, both shallow and deep embeddings
of each language are given. The modularity of our formalizations allows
proofs and theorems to be reused across different embeddings. We illus-
trate the advantages of our modular formalizations by proving the mutual
exclusion property of the Bakery algorithm in different embeddings.

1 Introduction

Verifying systems in proof assistants consists of several tedious tasks. Firstly,
system behavior has to be specified. Issues such as finite versus infinite states,
deterministic versus non-deterministic computation, interleaving versus concur-
rent semantics need be resolved in behavioral specification. Secondly, system
requirements are needed. Formalisms featuring linear- or branching-time seman-
tics, logical or algebraic descriptions are available. Finally, the verification need
be carried out in proof assistants to show that system behavior is indeed ex-
pected.

Since describing system behavior and requirements is very tedious and error-
prone, model and requirement specification languages are hence developed to
help verifiers specify system behavior and requirements. Several specification
languages have also been formalized in proof assistants. Roughly, two kinds of
formalizations are possible. A shallow embedding defines the semantics of spec-
ification languages in logics of proof assistants. Alternatively, a deep embedding
additionally encodes syntactic representations of specification languages in proof
assistants. They have consequently different characteristics.

Shallow embeddings are easier to carry out since syntactic representations
are not defined. But structural induction on specification languages is not pos-
sible due to the lack of syntactic representations. Deep embedding, on the other

hand, are more complicated. Nevertheless, their syntactic representations allow
to analyze properties about encoded languages. In practice, objectives of formal-
izations determine which embedding is used. If a specific requirement is to be
verified on a particular system, shallow embeddings of both model and require-
ment specification languages are preferred, for instance.

For some applications, switching between different embeddings may be nec-
essary. If a verifier is developing a new model specification language, she may
try some examples in shallow embedding at early stages of the development.
When the language is more mature, deep embedding will be necessary to prove
properties about the language itself. Ideally, the theorems and proofs in different
embeddings should be reused. But it is not clear how the migration can be done
easily.

In this paper, we present modular formalizations of model and requirement
specification languages in the proof assistant Coq [4]. We give the shallow and
deep embeddings of both specification languages and compare their formaliza-
tions in different embeddings. More importantly, our modular formalizations
allow to share theorems and proofs between embeddings. We have developed
tactics for different embeddings to reduce the migration cost. As an example,
we prove the mutual exclusion of the Bakery algorithm with indefinite number
of processes in both embeddings.

We formalize Reactive Modules [1] for system descriptions. Reactive Modules
is a model specification language for synchronous, shared-memory, and concur-
rent processes. Multiple processes may update different variables simultaneously.
The temporal logic CTL∗ is used for requirement specifications. Unlike linear-
time temporal logic (LTL) and computation tree logic (CTL), CTL∗ admits
both path and state formulae and is therefore more expressive than LTL and
CTL. Purely path- or state-based semantics would not suffice for CTL∗. Our
formalization integrates both semantics seamlessly.

The branching-time temporal logic µ-calculus has been formalized in Coq [10,
16], LEGO [18], and ACL2 [8] with different intentions. A formalization of LTL
can be found in Coq [3]. The temporal logic of actions TLA [5] has been for-
malized in Isabelle [9]. A shallow embedding of CTL can be found in [2]. Since
CTL∗ is an accessible generalization of LTL and CTL, our formalization sub-
sumes prior formalizations of LTL and CTL. Furthermore, verifiers are more
familiar with CTL∗ than µ-calculus. Our formalization would be more useful to
them.

In [14, 13, 11, 12], I/O automata [6] and (extensions to) linear-time temporal
logic [7] are formalized in Isabelle [15]. An I/O automaton is formalized as the
set of admissible paths. A path in turn is formalized as a lazy list or a function
on natural numbers. The authors are able to prove meta-theorems about I/O
automata in their framework. Our formalization of Reactive Modules is more
syntactic. We provide constants to help verifiers define their models and give
corresponding syntactic representation. Verifiers will find it more accessible than
pure semantic-based formalizations.



The paper is organized as follows. Section 2 gives the preliminaries. Both the
shallow and deep embeddings of Reactive Modules are presented in Section 3.
The embeddings of CTL∗ are shown in Section 4. The verification of the Bakery
algorithm is discussed in Section 5. Finally, Section 6 discusses possible future
work and concludes the paper.

2 Preliminaries

Let AP be the set of atomic propositions. A Kripke structure K is a 4-tuple
(S, S0, δ, L) where S is the set of states, S0 ⊆ S the set of initial states, δ ⊆ S×S
the total transition relation, and L : S → 2AP the labeling function which assigns
atomic propositions to each state.

Given a set of atomic propositions AP , a state formula f is defined induc-
tively as follows.

– If p ∈ AP , then p is a state formula;
– If f and f ′ are state formulae, then ¬f , f ∨f ′, and f ∧f ′ are state formulae;
– If g is a path formula, then Eg and Ag are state formulae.

A path formula g, on the other hand, is constructed by the following rules.

– If f is a state formula, then f is also a path formula;
– If g and g′ are path formulae, then ¬g, g∨ g′, g∧ g′, Xg, Fg, Gg, gUg′, and

gWg′ are path formulae.

A formula in the full branching-time temporal logic CTL∗ is either a state or a
path formula.

Let K = (S, S0, δ, L) be a Kripke structure. A path is an infinite sequence of
states π = s0s1 · · · sn · · · such that (si, si+1) ∈ δ for i ≥ 0. We define π(i) = si

and the i-th suffix πi = sisi+1 · · · sn · · · for any natural number i. Let s ∈ S.
Define the relations K, s |= f and K, π |= g as follows.

K, s |= p if p ∈ L(s)
K, s |= ¬f if K, s 6|= f
K, s |= f ∨ f ′ if K, s |= f or K, s |= f ′

K, s |= Eg if there is a path π with π(0) = s such that K, π |= g
K, π |= f if K, π(0) |= f
K, π |= ¬g if K, π 6|= g
K, π |= g ∨ g′ if K, π |= g or K, π |= g′

K, π |= Xg if K, π1 |= g
K, π |= gUg′ if there exists an i ≥ 0 such that K, πi |= g′ and

K, πj |= g for all 0 ≤ j < i

The semantics of other operators is obtained by φ ∧ φ′ ≡ ¬(¬φ ∨ ¬φ′), Af ≡
¬E¬f , Fg ≡ trueUg, Gg ≡ ¬F¬g, and gWg′ ≡ gUg′ ∨Gg. Moreover, we say
the state s in K satisfies the state formula f if K, s |= f . Similarly, the path π
satisfies the path formula g if K, π |= g.



A reactive module consists of a set of atoms that define how variables are
updated. Both the old and new values of a variable can be referred in atoms.
If a variable x is updated by an atom A, we say x is controlled by A. If its old
value is referred in A, we say it is read by A. If its new value is referred, we say
x is awaited by A. An atom is a 5-tuple A = (V A

c , V A
r , V A

a , CA
i , CA

u ) where V A
c

is the set of variables controlled by A, V A
r the set of variables read by A, V A

a

the set of variables awaited by A, CA
i the set of initial guarded commands, and

CA
u the set of updating guarded commands.

Variables are updated by their controlling atoms in each step. In the begin-
ning of each step, external and uncontrolled variables are assigned to arbitrary
values nondeterministically. An atom updates its controlled variables only after
all its awaited variables have been updated in the same step.

We write x and x′ for the old and new values of the variable x respec-
tively. For example, consider the following textual representation of an atom A
= (V A

c , V A
r , V A

a , CA
i , CA

u ) where V A
c = {x, y}, V A

r = {x, y, z}, and V A
a = {z}.

atom A controls x y reads x y z awaits z
inits

[] true → x := 1; y := 2
[] true → x := 2; y := 1

updates
[] x = 1 → x’ := y; y’ := z’
[] ¬(x = 1) → x’ := z; y’ := x

There are two initial guarded commands in CA
i and two updating guarded com-

mands in CA
u . One of the updating guarded command, for example, assigns the

old value of y to x and the new value of z to y when the old value of x is 1.

3 Reactive Modules

Our formalization differs from other formalizations in its modularity. Instead of
formalizing the requirement specification language on top of the model specifica-
tion language, an abstraction of Kripke structures is introduced as the interfaces
in our formalization. Any formalization of Reactive Modules must conform to the
abstract Kripke interface. In particular, both the shallow and deep embeddings
share the same interface so that they can be used in our CTL∗ formalization
interchangeably.

We formalize the abstraction as a Coq module type. The following fragment
shows the abstract Kripke interface.

Module Type KRIPKE .

Parameters Var St : Set .

Parameter Val : St -> Var -> nat .

Parameter Init : St -> Prop .

Parameter Succ : St -> St -> Prop .

End KRIPKE .



In the module type KRIPKE, the variable set Var and the state set St are declared.
The valuation of each variable x on a given state s is obtained by Val s x. A
state s is an initial state if Init s holds. Similarly, the state t is a successor of
s if Succ s t holds. The labeling function L is not declared for it is formalized
as state predicates of type St -> Prop.

In the shallow embedding, only the semantics of Reactive Modules is en-
coded. Constants are provided to help verifiers construct reactive modules. New
constructs can be formalized by adding new constants. On the other hand, both
the syntax and semantics are encoded in the deep embedding. Each syntactic
class of Reactive Modules corresponds to an inductive type in Coq. A seman-
tic function is defined to give the meaning of constructs in each syntactic class.
Adding a new construct requires modifying the inductive type and the semantic
function of its syntactic class.

3.1 Shallow embedding

In the shallow embedding SRM, variables in reactive modules are of type Var. A
state is a function which maps variables to values. For simplicity, all variables
in our formalization have values in natural numbers (nat).

Variable Var : Set .

Definition St : Set := Var -> nat .

Since both the old and new values of variables may be referred in reactive
modules, an expression is formalized as a function of type St -> St -> nat where
the arguments are the current and the next state respectively. It returns a nat

term as its semantic value. A Boolean expression, on the other hand, is a predi-
cate on the current and next states.

Definition Exp : Set := St -> St -> nat .

Definition BExp : Type := St -> St -> Prop .

It is easy to encode constructs in SRM. For instance, the following definition
defines the constant add for the arithmetic addition operator.

Definition add (e1 e2 : Exp) : Exp := fun s s’ : St => e1 s s’ + e2 s s’ .

The constant add passes the current and next states to its parameters, evaluates
the subexpressions, and finally returns the sum as its result. Other constants for
expressions are defined similarly.

Likewise, constants for Boolean expressions are given. For example, the fol-
lowing constants define the truth value true and the negation.

Definition wahr : BExp := fun s s’ : St => True .

Definition nicht (be : BExp) : BExp := fun s s’ : St => ~ be s s’ .

Another constant which is handy in the specification of mutual exclusion is
forall nat. It checks whether the operand f of type nat -> BExp holds for all
nat.

Definition forall_nat (f : nat -> BExp) : BExp :=

fun s s’ : St => forall n : nat, f n s s’ .



A command in CA
u is a predicate over the current and next states while an

initial command in CA
i a predicate over states.

Definition Command : Type := St -> St -> Prop .

Definition ICommand : Type := St -> Prop .

Predefined constants are available for commands. For example, Assign assigns
an expression to a variable, and Comp makes a new command out of two. Similar
constants in ICommand can be defined as well.

Definition Assign (v : Var) (exp : Exp) : Command :=

fun s s’ : St => s’ v = exp s s’ .

Definition Comp (c c’ : Command) : Command :=

fun s s’ : St => c s s’ /\ c’ s s’ .

Note that the composition of commands is formalized as conjunction. It conforms
to the semantics defined in [1].

3.2 Deep embedding

The syntax and semantics are defined explicitly in the deep embedding DRM. For
each syntactic class, we define an inductive type for its terms and a semantic
function for its semantics.

Syntax. For expressions, we define the inductive type Exp as follows.

Inductive Exp : Set :=

| add : Exp -> Exp -> Exp

...

The inductive type BExp for Boolean expressions is defined similarly.

Inductive BExp : Set :=

| wahr : BExp

| nicht : BExp -> BExp

| forall_nat : (nat -> BExp) -> BExp

...

In DRM, elements of Exp and BExp are constructed inductively. Subsequently,
properties about expressions or Boolean expressions can be proved by induc-
tion. On the other hand, Exp and BExp are of functional types in the shallow
embedding. Inductive reasoning about these expressions would not work in SRM.

A command can be an assignment or a composition of commands. A guarded
command is a record which contains a Boolean expression and a command.
Finally, an atom is a record of a list of controlled variables, a list of initial
guarded commands, and a list of updating guarded commands.

Inductive Command : Set :=

| assign (v : Var) (exp : Exp) | comp (c c’ : Command) .

Record GuardedCommand : Set := mkGuardedCommand

{ guard : BExp; command : Command } .

Record Atom : Set := mkAtom

{ controls : list Var; inits : list GuardedCommand;

updates : list GuardedCommand } .



Semantics. To give a term its meaning, a semantic function for its syntactic
class is defined. The semantic functions eval and beval compute the semantic
value of any Exp and BExp term respectively based on the current and next states.

Fixpoint eval (exp : Exp) (s s’ : St) { struct exp } : nat :=

match exp with

| add exp’ exp’’ => (eval exp’ s s’) + (eval exp’’ s s’)

...

end .

Fixpoint beval (bexp : BExp) (s s’ : St) { struct bexp } : Prop :=

match bexp with

| wahr => True

| nicht bexp’ => ~ (beval bexp’ s s’)

| forall_nat F => forall n : nat, beval (F n) s s’

...

end .

Compare the definition of Exp in SRM and the type of the semantic function
eval in DRM.

Exp := St -> St -> nat (* in SRM *)

eval : Exp -> St -> St -> nat (* in DRM *)

The type of the term eval add in DRM corresponds to the type of the constant
add in SRM. Indeed, the semantic function eval collects the corresponding def-
initions of Exp constants in the shallow embedding. The correspondence will
be observed throughout the formalizations. It helps us differentiate embeddings
systematically.

The semantic function Exec for commands can be defined easily.

Fixpoint Exec (c : Command) (s s’ : St) { struct c } : Prop :=

match c with

| assign v exp => s’ v = (eval exp s s’)

| comp c’ c’’ => Exec c’ s s’ /\ Exec c’’ s s’

end .

In order to define the parameter Succ in the module type KRIPKE, several
auxiliary definitions are needed. Given a term of Atom and two states s, s’,
SuccAtom holds if s’ can be obtained from s by executing an enabled guarded
command. Hence, the state s’ is a successor of s if s’ can be obtained from s

by executing enabled guarded commands of all atoms simultaneously.

Definition SuccAtom (atom : Atom) (s s’ : St) : Prop :=

exists gcmd : GuardedCommand, In gcmd (updates atom) /\

beval (guard gcmd) s s’ /\ Exec (command gcmd) s s’ .

Definition Succ (atoms : Ensemble Atom) (s s’ : St) : Prop :=

forall atom : Atom, In Atom atoms atom -> SuccAtom atom s s’ .

4 CTL∗

As in the formalizations of Reactive Modules, the shallow and deep embed-
dings of CTL∗ must conform to the same interface. Moreover, we would like to



maximize the applicability of our CTL∗ formalization, and therefore would not
commit to particular model specification languages. CTL∗ is therefore formalized
as a functor on modules of type KRIPKE.

In the abstract interface of CTL∗, the formalization of paths is required.
Standard functions such as hd, tl, and cons are declared.

Module Type CTLS (K : KRIPKE) .

Parameter Path : Set .

Parameter hd : Path -> K.St .

Parameter tl : Path -> Path .

Parameter cons : forall (s : K.St) (pi : Path),

K.Succ s (hd pi) -> Path .

Parameter suffix : nat -> Path -> Path .

...

The function suffix i pi returns the i-th suffix of the path pi. It is used in the
formalizations of various temporal operators.

The declarations of state and path formulae follows those of paths. The sat-
isfaction relations K, s |= f and K, π |= g will be formalized by model s and
model p respectively.

Parameters state_formula path_formula : Type .

Parameter model_s : state_formula -> K.St -> Prop .

Parameter model_p : path_formula -> Path -> Prop .

Parameter A : path_formula -> state_formula .

Parameter G : path_formula -> path_formula .

Axiom def_A : forall (s : K.St) (f : path_formula),

model_s s (A f) <-> forall pi : Path, s = hd pi -> model_p pi f .

...

End CTLS .

The types of path quantifiers and temporal operators are defined. For instance,
the path quantifier A must be of type path formula -> state formula. Finally,
semantic constraints on temporal operators are explicated.

4.1 Paths

Paths are formalized by imposing constraints on streams. A stream is an infinite
sequence of states. A stream of states σ = s0s1 · · · sn · · · is a path if and only
if si+1 is a successor of si for all i ≥ 0. For simplicity, we formalize streams as
functions from natural numbers to states in both embeddings of CTL∗. Other
formalizations via domain theory [13] or coinductive types [17] are possible.

Definition stream : Set := nat -> K.St .

Definition is_path (str : stream) : Prop :=

forall i : nat, K.Succ (str i) (str (i + 1)) .

Definition Path : Set := { pi : stream | is_path pi } .

The predicate is path checks whether successive states in the given stream
satisfy the transition relation K.Succ. Path is hence a set of streams satisfying the
predicate is path. Auxiliary functions such as hd, tl, and cons can be defined
easily and are skipped here.



4.2 Shallow embedding

State and path formulae are formalized as predicates over states and paths in
the shallow embedding SCTLS respectively.

Definition path_formula : Type := Path -> Prop .

Definition state_formula : Type := St -> Prop .

Consider, for example, the path quantifier A. It makes a state formula out
of a path formula. In our formalization, it transforms a path predicate to a state
predicate.

Definition A (P : path_formula) : state_formula :=

fun s : St => forall pi : Path, s = hd pi -> P pi .

Similarly, the temporal operator G checks whether all suffixes (suffix i pi)
of the input path (pi) satisfy the given path predicate.

Definition G (P : path_formula) : path_formula :=

fun pi : Path => forall i : nat, P (suffix i pi) .

4.3 Deep embedding

The deep embedding DCTLS follows a similar style in Reactive Modules. We de-
fine two inductive types for the syntactic classes of path and state formulae
respectively. Semantic functions for the syntactic classes are defined accordingly.

Syntax. The syntactic classes of state and path formulae correspond to the
mutually inductive types state formula and path formula respectively.

Inductive state_formula : Type :=

| A : path_formula -> state_formula

...

with path_formula : Type :=

| X : path_formula -> path_formula

| G : path_formula -> path_formula

...

Semantics. The semantics of state and path formulae are defined mutually
recursively by the functions model s and model p respectively.

Fixpoint model_s (f : state_formula) (s : St) {struct f} : Prop :=

match f with

| A g => forall pi : Path, s = hd pi -> model_p g pi

...

end

with model_p (g : path_formula) (pi : Path) {struct g} : Prop :=

match g with

| X g’ => model_p g’ (tl pi)

| G g’ => forall k : nat, model_p g’ (suffix k pi)

...

end .



The correspondence of both embeddings can be re-illustrated.

(* in SCTLS *)

state_formula := St -> Prop

path_formula := Path -> Prop

(* in DCTLS *)

model_s : state_formula -> St -> Prop

model_p : path_formula -> Path -> Prop

The semantic functions model s and model p in DCTLS collect the definitions of
corresponding operators in SCTLS. As in the formalization of Reactive Modules, it
is easy to add new operators in the shallow embedding. But DCTLS allows verifiers
to analyze CTL∗ formulae inductively.

5 The Bakery Algorithm

In this section, we demonstrate the ease of proof sharing in our modular for-
malization. The mutual exclusion of the Bakery algorithm will be used as the
example. More precisely, three versions of the property are proved in different
embeddings. We first specify the Bakery algorithm with the shallow and deep
embeddings of Reactive Modules and obtain the modules SBakery and DBakery

respectively. Two versions of mutual exclusion are specified by the shallow and
deep embeddings of CTL∗ as SMUTEX and DMUTEX respectively. The similarity be-
tween the proofs of SMUTEX in SBakery and DBakery will be illustrated. Finally,
the theorem SMUTEX is reused to prove DMUTEX in DBakery (Figure 1).

SBakery SMUTEX

DBakery DMUTEX

-|=

����*|=

-
|=

Fig. 1. Different Versions of Mutual Exclusion

We deliberately use the same notation in both embeddings to demonstrate
the similarity of proofs. Hence some functions and constructors may have dif-
ferent types in different embeddings. For example, the term or is a function in
SBakery but a constructor in DBakery. The following constants are used in both
encodings.

Definition reqCS := 1 .

Definition inCS := 2 .

Definition outCS := 3 .

Parameter n : nat .

Axiom n_gt_0 : n > 0 .

Parameters pc y : nat -> Var .



Intuitively, n denotes the number of processes. pc i is the control location of the
process i, which can be reqCS, inCS, or outCS. y i is the ticket number of the
process i. Then the condition that process i can enter the critical section is as
follows.

Definition cond (i : nat) : BExp :=

let cond_i_j (i : nat) (j : nat) : BExp :=

y j = 0 \/ y i < y j \/ (y i = y j /\ i < j) in

forall_nat j, i <> j --> cond_i_j i j .

It states that if all other process j has ticket number 0, a larger ticket number (y
i < y j), or the index j greater than i when both have the same ticket number,
then the process i may go into the critical section. Note that cond has different
types in different embeddings.

5.1 Shallow modeling

In the shallow modeling, the initial and updating commands are defined as pred-
icates. To mimic the textual representation of Reactive Modules, several nota-
tions are defined. For instance, x′ := e stands for Assign x e; Updates [] gcmd0

[] gcmd1 · · · stands for fun s s’ : St => gcmd0 s s’ \/ gcmd1 s s’ \/ · · ·, and
so on.

Definition atom_inits (i : nat) : ICommand :=

(Inits

[] wahr -> (pc i) := outCS; (y i) := 0) .

Definition atom_updates (i : nat) : Command :=

(Updates

[] pc i = outCS -> (pc i)’ := reqCS; (y i)’ := max_y + 1

[] pc i = reqCS /\ cond i -> (pc i)’ := inCS; (y i)’ := y i

[] pc i = reqCS /\ ~ cond i -> (pc i)’ := reqCS; (y i)’ := y i

[] pc i = inCS -> (pc i)’ := outCS; (y i)’ := 0) .

Definition sys_inits (s : St) : Prop :=

forall (i : nat), i < n -> atom_inits i s .

Definition sys_updates (s s’ : St) : Prop :=

forall (i : nat), i < n -> atom_updates i s s’ .

The concrete module SBakery of type KRIPKE is created as follows.

Module SBakery <: KRIPKE .

Definition Var := SRM.Var .

Definition St := SRM.St .

Definition Val (s : St) (v : Var) := s v .

Definition Init := sys_inits .

Definition Succ := sys_updates .

End SBakery .

With the module SBakery in place, we can instantiate the module SBakerySCTL

as follows.

Module SBakerySCTL := SCTLS SBakery .



Mutual exclusion is expressed as a state formula smutex.

Definition smutex : state_formula :=

fun s : St => forall i j : nat, i < n -> j < n -> i <> j ->

s (pc i) <> inCS \/ s (pc j) <> inCS .

It is now straightforward to prove the following lemma in SBakerySCTL.

Lemma SMUTEX : forall s : St, Init s -> model_s (A (G smutex)) s .

5.2 Deep modeling

To model the Bakery algorithm in the deep embedding, we define the Atom term
atom i for each natural number i. The notations x′ := e and Updates [] gcmd0

[] gcmd1 · · · stand for assign x e and [ gcmd0 ; gcmd1 ; · · · ] respectively.

Definition atom (i : nat) : Atom := mkAtom (pc i::y i::nil)

(Inits

[] wahr -> (pc i) := outCS; (y i) := 0)

(Updates

[] pc i = outCS -> (pc i)’ := reqCS; (y i)’ := max_y + 1

[] pc i = reqCS /\ cond i -> (pc i)’ := inCS; (y i)’ := y i

[] pc i = reqCS /\ ~ cond i -> (pc i)’ := reqCS; (y i)’ := y i

[] pc i = inCS -> (pc i)’ := outCS; (y i)’ := 0) .

The system with n atoms is defined as an ensemble of atoms (Ensemble Atom).

Definition sys : Ensemble Atom :=

fun a : Atom => exists i : nat, i < n /\ a = atom i .

The concrete model of type KRIPKE is defined as follows.

Module DBakery <: KRIPKE .

Definition Var := DRM.Var .

Definition St := DRM.St .

Definition Val (s : St) (v : Var) := s v .

Definition Init := DRM.Init sys .

Definition Succ := DRM.Succ sys .

End DBakery .

Similar to using shallow embedding, we instantiate the module DBakerySCTL.

Module DBakerySCTL := SCTLS DBakery .

The following lemma shows the mutual exclusion in DBakerySCTL.

Lemma SMUTEX : forall s : St, Init s -> model_s (A (G smutex)) s .

Recall that both SBakerySCTL.SMUTEX and DBakerySCTL.SMUTEX specify the mu-
tual exclusion property in the shallow embedding of CTL∗. Further, SBakery

and DBakery are merely two encodings of the Bakery algorithm in different em-
beddings. Intuitively, the proofs of SBakerySCTL.SMUTEX and DBakerySCTL.SMUTEX

should be similar. If so, one may establish DBakerySCTL.SMUTEX by adopting the



in SBakerySCTL.SMUTEX in DBakerySCTL.SMUTEX

gen cmd Hsucc i Hi unfold cond i j; gen cmd Hsucc (atom i);

gen cmd Hsucc j Hj unfold cond i j; gen cmd Hsucc (atom j);

clear Hsucc. clear Hsucc.
· · · · · ·
case (eq nat dec (s (pc j)) inCS); intro . case (eq nat dec (s (pc j)) inCS); intro .

generalize (Hin j Hj e); clear Hin; intro . generalize (Hin j Hj e); clear Hin; intro .
generalize (Hreq i Hi H0); clear Hreq; generalize (Hreq i Hi H0); clear Hreq;

intro . intro .
generalize generalize

(Hin’ j Hj e i Hi (sym not eq Hij)); (Hin’ j Hj e i Hi (sym not eq Hij));
clear Hin’; intro . clear Hin’; intro .

generalize (H1 j Hij); clear H1 . generalize (H1 j Hij); clear H1 .
intros; elim all or; elim all and; intros; elim all or; elim all and;

auto contradict . auto contradict .

auto replace H; auto . auto replace H4; auto .

Fig. 2. Proof Scripts

proof of SBakerySCTL.SMUTEX. The cost of migrating from SBakery to DBakery

would be greatly reduced.
It is indeed the case in this example. Figure 2 shows fragments of the proofs of

SBakerySCTL.SMUTEX and DBakerySCTL.SMUTEX, where differences are underlined.
In the beginning of the proof, the tactic gen cmd generates relations between the
current and next states according to the transition relation Hsucc specified by
the atoms. For example, for the guarded command [] pc i = inCS -> (pc i)’

:= outCS; (y i)’ := 0 in atom i, gen cmd would generate two assumptions H0

: s (pc i) = inCS and H1 : s’ (pc i) = outCS /\ s’ (y i) = 0 in the context
where the guarded command is fired. We provide two versions of the tactic for
different embeddings to help verifiers migrate their proofs. The remaining frag-
ments analyze the scenario where process i enters the critical section and process
j stutters. The only difference is due to the names generated during the case
analysis. The tactics elim all or, elim all and, auto contradict, and auto replace
simplify assumptions and look for contradiction. They are shared in both em-
beddings.

Figure 2 shows that the proofs of SBakerySCTL.SMUTEX and DBakerySCTL.SMUTEX

are very similar despite using different encodings. In both proofs, the tactic
gen cmd translates the transition relation into logical relations between the cur-
rent and next states. The proofs do not involve any encoding henceforth. If
verifiers follow a similar proof strategy, the proof within the shallow embedding
could be transformed to a proof within the deep embedding, and the migration
cost is therefore reduced.

It is even simpler to prove the mutual exclusion property with the deep em-
bedding of CTL∗ from DBakerySCTL.SMUTEX. We instantiate the module DBakeryDCTL
of the deep embedding.



Module DBakeryDCTL := DCTLS DBakery .

The atomic proposition smutex in the shallow embedding becomes the fol-
lowing term in the deep embedding of CTL∗.

Definition dmutex : AP :=

forall_nat (fun i : nat => forall_nat (fun j : nat =>

i < n --> j < n --> i <> j -->

(pc i) <> inCS \/ (pc j) <> inCS)) .

The property can now be specified with the module DBakeryDCTL as follows.

Lemma DMUTEX : forall s : St, Init s -> model_s (A (G dmutex)) s .

To prove DMUTEX, note that both DBakerySCTL.SMUTEX and DBakeryDCTL.DMUTEX

are of the same type in CTLS. A simple application of DBakerySCTL.SMUTEX proves
DBakeryDCTL.DMUTEX trivially.

6 Future Work and Conclusion

Migration from one formalization to another requires effort, even for different
embeddings of the same formalism. In this paper, we exploit the Coq module
system and give both the shallow and deep embeddings of two specification
languages. We show how the Bakery algorithm is specified in both embeddings of
Reactive Modules, as well as the mutual exclusion in both embeddings of CTL∗.
Three versions of the property are formally proved as examples. The examples
suggest that the migration cost can be alleviated by the modular formalization
presented in the paper.

Our formalization of the Reactive Modules could still be improved. In the
future, we would also like to formalize other model specification languages within
our framework. Moreover, the integration with different model checkers may be
possible by formalizing model specification languages of different model checkers.
Other formalizations of the requirement specification languages can also be done.
In particular, we are working on a comparison of different formalizations of
CTL∗.
Acknowledgment. We would like to thank Yih-Kuen Tsay and anonymous
reviewers for their constructive comments in improving the paper.

References

1. Alur, R., Henzinger, T.: Reactive modules. In: Proceedings of the 11th IEEE
Symposium on Logic in Computer Science. (1996) 207–218

2. Bauer, G.: Some properties of CTL. Technische Universität München. (2001)
Isabelle/Isar document.

3. Coupet-Grimal, S.: An axiomatization of linear temporal logic in the calculus of
inductive constructions. Logic and Computation 13(6) (2003) 801–813

4. Huet, G., Kahn, G., Paulin-Mohring: The Coq proof assistant: a tutorial: version
6.1. Technical Report 204, Institut National de Recherche en Informatique et en
Automatique (1997)



5. Lamport, L.: The temporal logic of actions. ACM Transactions on Programming
Languages and Systems 16(3) (1994) 872–923

6. Lynch, N.: Distributed algorithms. Morgan Kaufmann (1996)
7. Manna, Z., Pnueli, A.: The temporal framework for concurrent programs. In Boyer,

R., Moore, J., eds.: The Correctness Problem in Computer Science. Academic Press
(1981) 215–274

8. Manolios, P.: Mu-calculus model-checking. In: Computer-Aided Reasoning: ACL2
Case Studies. Kluwer Academic Publishers (2000) 93–111

9. Merz, S.: Isabelle/TLA. Technische Universität München. (1998) Isabelle/Isar
document.

10. Miculan, M.: On the formalization of the modal µ-calculus in the calculus of
inductive constructions. Information and Computation 164(1) (2001) 199–231

11. Müller, O.: I/O automata and beyond - temporal logic and abstraction in Isabelle.
In Grundy, J., Newey, M., eds.: Theorem Proving in Higher Order Logics. Volume
1479 of LNCS., Springer-Verlag (1998) 331–348

12. Müller, O.: A Verification Environment for I/O Automata Based on Formalized
Meta-Theory. PhD thesis, Technische Universität München (1998)

13. Müller, O., Nipkow, T.: Traces of I/O automata in Isabelle/HOLCF. In Bidoit, M.,
Dauchet, M., eds.: Theory and Practice of Software Development. Volume 1214 of
LNCS., Springer-Verlag (1997) 580–595

14. Nipkow, T., Slind, K.: I/O automata in Isabelle/HOL. In Dybjer, P., Nordström,
B., Smith, J., eds.: Types for Proofs and Programs. Volume 996 of LNCS., Springer-
Verlag (1995) 101–119

15. Paulson, L.C., Nipkow, T.: Isabelle tutorial and user’s manual. Technical Report
TR-189, Computer Laboratory, University of Cambridge (1990)

16. Sprenger, C.: A verified model checker for the modal µ-calculus in Coq. In Steffen,
B., ed.: TACAS. Volume 1384 of LNCS., Springer-Verlag (1998) 167–183

17. Tsai, M.H., Wang, B.Y.: Formalization of CTL∗ in calculus of inductive construc-
tions. submitted for publication (2006)

18. Yu, S., Luo, Z.: Implementing a model checker for LEGO. In Fitzgerald, J., Jones,
C.B., Lucas, P., eds.: Formal Methods Europe. Volume 1313 of LNCS., Springer-
Verlag (1997) 442–458


